SUPPORTING FILE FOR OUR PAPER "BIRATIONAL GEOMETRY AND THE CANONICAL RING OF A FAMILY OF DETERMINANTAL 3-FOLDS"

VLADIMIR LAZIĆ AND FRANK-OLAF SCHREYER

ABSTRACT. In the auxiliar file we list the input and output of Macaulay2 computations used in the proof and in the study of an example of our paper "Birational geometry and the canonical ring of a family of determinantal 3-folds".

CONTENTS

1. APPENDIX: SUPPORTING COMPUTATIONS IN MACAULAY2

We use the Computer algebra system Macaulay2, in particular the package "TateOnProducts" [1] and the file [2].

1.1. A specific example. We analyse a specific example for $b = 1$ over a finite field to get some idea of the situation.

```
i1 : b=1;
i2 : kk=ZZ/nextPrime 10^3; -- a finite ground field
i3 : P1=kk[z_0,z_1];
i4 : P2=kk[x_0..x_2];
i5 : P3=kk[y_0..y_3];
i6 : P2xP3=P2**P3;
i7 : P1xP2xP3=P1**P2xP3;
i8 : P1xP3=P1**P3; -- various coordinate rings
i9 : sA= (syz gens ideal sub(vars P2,P2xP3))**
        P2xP3^{{1,0}}++id_(P2xP3^{1:{0,b-1}})
i10 : G=image sA -- the bundle G
o10 = image | -x_1 0    -x_2 0 |
            | x_0  -x_2 0    0 |
            | 0    x_1  x_0  0 |
            | 0    0    0    1 |
i11 : phi=sA*random(source sA,P2xP3^{2:{-1,-1}});
i12 : m=transpose phi;
```

```
i13 : minimalBetti coker m
               0 1 2
o13 = total: 2 4 2
          -2: 2 . .
          -1: . 4 1
           0: . . .
           1: . . 1
o13 : BettiTally
i14 : betti (IX=ann coker m)
               0 1
o14 = total: 1 4
           0: 1 .
           1: . .
           2: . 1
           3: . 3
i15 : codim IX
o15 = 2
i16 : betti(fIX=res IX)
               0 1 2
o16 = total: 1 4 3
           0: 1 . .
           1: . . .
           2: . 1 .
           3: . 3 3
i17 : I=ideal (sub(vars P1,P1xP2xP3)*sub(m,P1xP2xP3));
i18 : tally degrees source gens I
o18 = Tally{{1, 1, 1} => 4}
```

We computed the ideal I_X of $X \subset \mathbb{P}^2 \times \mathbb{P}^3$ and I of X in $\mathbb{P}1 \times \mathbb{P}^2 \times \mathbb{P}3$. Next we compute the ideal I_{X^1} of the image in $\mathbb{P}^1 \times \mathbb{P}^3$.

```
i19 : IX1=ideal mingens sub(saturate(I,
      ideal sub(vars P2,P1xP2xP3)),P1xP3);
i20 : degrees source gens IX1
o20 = {{2, 2}}
```

X^1 is a hypersurface of bi-degree $(2,2)$. The determinant of its hessian is the branch divisor of a double cover $Y \to \mathbb{P}^3$.

```
i21 : hess=diff((vars P1xP3)_{0,1},transpose
                      diff((vars P1xP3)_{0,1},gens IX1));
i22 : B=ideal det sub(hess,P3);
i23 : degree B
o23 = 4
```

Next we compute image E of the exceptional locus of $X \to X^1$. It is the loci where the 3×4 matrix below drops rank.

```
i24 : fib=minors(2,sub(contract(transpose
               sub(vars P2,P1xP2xP3),gens I),P1xP3));
```

```
i25 : E=saturate(saturate(fib,sub(ideal vars P1,P1xP3)),
                 sub(ideal vars P3,P1xP3));
i26 : minimalBetti E
                 0 1 2 3
o26 = total: 1 6 8 3
          0: 1 . . .
          1: . 6 8 3
i27 : C=trim sub(E,P3);
i28 : dim C, degree C, genus C, betti res C
                          0 1 2
o28 = (2, 3, 0, total: 1 3 2)
                    0: 1 . .
                    1: . 3 2
i29 : cX1=decompose (sub(C,P1xP3)+IX1);
i30 : apply(cX1,c->(dim c,minimalBetti c))
                     0 1 2 3              0 1 2 3
o30 = {(3, total: 1 6 8 3), (3, total: 1 5 5 1)}
               0: 1 . . .           0: 1 . . .
               1: . 6 8 3           1: . 3 2 .
                                    2: . 2 3 .
                                    3: . . . 1
```

The image of E in $\mathbb{P}^3$ is a rational normal curve C. Its preimage in X^1 has two components, one of which is blown-up by the map $X \to X^1$:

```
i31 : cX=decompose (sub(C,P2xP3)+IX);
i32 : apply(cX,c->(dim c,minimalBetti c))
                     0 1 2 3              0  1  2  3 4
o32 = {(4, total: 1 6 8 3), (3, total: 1 10 20 15 4)}
               0: 1 . . .           0: 1  .  .  . .
               1: . 3 2 .           1: . 10 20 15 4
               2: . 3 6 3
```

Indeed the curve C intersect the branch divisor B tangentially in 6 points:

```
i33 : pts=C+B;
i34 : dim pts, degree pts, degree radical pts
o34 = (1, 12, 6)
```

We check that B has 8 A_1-singularities:

```
i35 : singB=ideal jacobian B;
i36 : singBr=radical singB;
i37 : singBr==ideal sub(hess,P3)
o37 = true
i38 : dim singB, degree singB, degree singBr
o38 = (1, 8, 8)
i39 : sub(singBr,P1xP3)+IX1==sub(singBr,P1xP3)
o39 = true
```

Thus the double cover $Y = V(w^2 - \det hess) \subset \mathbb{P}(1,1,1,1,b+1)$ has A_1 singularities as well, and $X^1 \to Y$ is a small resolutions of singularities.

Finally, we compute the cohomology matrix of $\mathcal{O}_X$ on $\mathbb{P}^2 \times \mathbb{P}^3$.

```
i40 : loadPackage("TateOnProducts")
i41 : (S,E)=productOfProjectiveSpaces({2,3},
      CoefficientField=>kk)
i42 : J=sub(IX,vars S);
i43 : SX=S^1/J;
i44 : cohomologyMatrix(SX,{-3,-5},{2,3})
o44 = | 10h   2    24   56         98   |
      | 4h    0    11   29         54   |
      | 0     0    4    12         24   |
      | 0     0    1    3          6    |
      | 6h3   2h3  0    0          2h   |
      | 20h3  8h3  h3   h2         2h   |
      | 44h3  20h3 4h3  4h2        4h2  |
      | 80h3  40h3 11h3 7h2        14h2 |
      | 130h3 70h3 24h3 2h3+10h2 26h2 |
```

1.2. **Verifying claims of the paper computationally.** We print the Macaulay2 input file of all needed computations, which we have decorated with some output as comments.

```
kk=QQ; -- the ground field
be=3; -- the degree of the forms b on P3
S=kk[x_0..x_2,y_0..y_3,b_(0,0)..b_(1,2),
     Degrees=>{3:{1,0},4:{0,1},6:{0,be}}];
-- the coordinate ring of P2xP3 with
-- in addition generic forms b_ij
y23=matrix apply(2,i->apply(3,j->y_(i+j)))
-*
o4 = | y_0 y_1 y_2 |
     | y_1 y_2 y_3 |
*-
-- the 2x3 matrix defining a rational normal curve in P3
kx=diagonalMatrix{1,-1,1}*
    (koszul(2,matrix{{x_0,-x_1,x_2}}))_{2,1,0}
-*
o5 = | 0    -x_2 x_1  |
     | x_2  0    -x_0 |
     | -x_1 x_0  0    |
*-
bb=matrix apply(2,i->apply(3,j->b_(i,j)))
-*
o6 = | b_(0,0) b_(0,1) b_(0,2) |
```

```
     | b_(1,0) b_(1,1) b_(1,2) |
*-
m=map(S^2,,(y23*kx|bb *transpose matrix{{x_0..x_2}}));
--  transpose m is the generic homomorphism 2O -> G
--  where G =ker(3O(1,1) ->O(2,1)) \oplus O(1,be)
betti (fm=res coker m)
-*
            0 1 2
o8 = total: 2 4 2
         0: 2 . .
         1: . 3 1
         2: . . .
         3: . 1 .
         4: . . .
         5: . . 1
*-
J=ann coker m;
betti res J
-*
             0 1
o10 = total: 1 4
          0: 1 .
          1: . .
          2: . 1
          3: . .
          4: . .
          5: . 3
*-
C=minors(2,y23);


-- We check Proposition 3.1
cF=decompose (J+C);
#cF==2
apply(cF,c->(codim c,betti res c))
-*
                  0 1 2 3               0  1  2  3 4
o14 = {(3, total: 1 6 8 3), (4, total: 1 10 20 15 4)}
               0: 1 . . .            0: 1  .  .  . .
               1: . 3 2 .            1: . 10 20 15 4
               2: . . . .
               3: . . . .
               4: . 3 6 3

*-
```

```
bby=diff(transpose basis({1,0},S), (gens cF_0)_{3..5});
bby-(transpose bb*matrix{{0,1},{-1,0}}*y23)==0
-- => the formulas in Prop. 3.1 (c) is correct
trim(minors(2,bby)+minors(2,y23))
-- => bby has has rank <= 1 over C
-- => E is a P^1-bundle over C.
yx25=map(S^2,,y23|matrix apply(2,i->apply(2,j->x_(i+j))))
-*
o18 = | y_0 y_1 y_2 x_0 x_1 |
      | y_1 y_2 y_3 x_1 x_2 |
*-
minors(2,yx25)==cF_1
--=> the formula in Prop. 3.1 (b) is correct

-- We check proposition 4.1:
P2xP3xP1=kk[x_0..x_2,y_0..y_3,b_(0,0)..b_(1,2),z_0,z_1,
    Degrees=>{3:{1,0,0},4:{0,1,0},6:{0,be,0},2:{0,0,1}}]
-- the coordinate ring of P2xP3xP1 with 6 generic
-- forms of degree (0,be,0) added
y23=sub(y23,P2xP3xP1);
kx=sub(kx,P2xP3xP1);
bb=sub(bb,P2xP3xP1);
J=ideal( matrix{{z_0,z_1}}*(y23*kx|bb
    *transpose basis({1,0,0},P2xP3xP1)));
-- the defining ideal in P2xP3xP1
betti J
-*
             0 1
o25 = total: 1 4
          0: 1 .
          1: . .
          2: . 3
          3: . .
          4: . 1
*-
N=diff(transpose basis({1,0,0},P2xP3xP1),gens J)
-*
o26 = | 0               y_2z_0+y_3z_1  -y_1z_0-y_2z_1
      | -y_2z_0-y_3z_1 0               y_0z_0+y_1z_1
      | y_1z_0+y_2z_1  -y_0z_0-y_1z_1 0
      ------------------------------------------
      b_(0,0)z_0+b_(1,0)z_1 |
      b_(0,1)z_0+b_(1,1)z_1 |
      b_(0,2)z_0+b_(1,2)z_1 |
*-
```

```
C=sub(C,P2xP3xP1);
cJC=decompose radical(J+C);
#cJC -- need to saturate
cJC1=apply(cJC,c->c:ideal basis({0,0,1},P2xP3xP1));
cJC2=apply(cJC1,c->c:ideal basis({1,0,0},P2xP3xP1));
cJC3=select(cJC2,c->not c==ideal (1_P2xP3xP1));
#cJC3==2
apply(cJC3,c->codim c)=={5,4}

C1=radical minors(2,N)+C;
-- C1 is the exceptional curve in P1xP3
-- of the map X_b -> X^1_b
yz24=y23|matrix{{-z_1},{z_0}}
-*
o36 = | y_0 y_1 y_2 -z_1 |
      | y_1 y_2 y_3 z_0  |
*-
minors(2,yz24)==C1
-- => the formula for C1 in Proposition 4.1 (b)
--    is correct

P1xP3=kk[z_0,z_1,y_0..y_3,b_(0,0)..b_(1,2),
    Degrees=>{2:{0,1},4:{1,0},6:{be,0}}]
N'=map(P1xP3^3,,sub(N,P1xP3));
J'=trim minors(3,N');
J1=radical J'
-- J1 defines the image X_b^1 of X_b in P1xP3
betti res J1
-*
             0 1
o42 = total: 1 1
          0: 1 .
          1: . .
          2: . .
          3: . .
          4: . .
          5: . 1
*-
f=J1_0;
M=map(P1xP3^2,,diff(transpose basis({0,1},P1xP3),
      diff(basis({0,1},P1xP3),gens J1)))
-- => the formula (8) is correct

C12=decompose (J1+sub(C,P1xP3));
C12_1== sub(C1,P1xP3)
```

```
-- => C1 is one of the components of the preimage
--    of C in X^1_b
C2=C12_0;
apply(C12,c->betti res c)
-*
              0 1 2 3         0 1 2 3
o48 = {total: 1 5 5 1, total: 1 6 8 3}
           0: 1 . . .      0: 1 . . .
           1: . 3 2 .      1: . 6 8 3
           2: . . . .
           3: . . . .
           4: . 2 3 .
           5: . . . 1
*-
(res C2).dd_2
-*
| -y_2 y_3  z_0b_(0,1)+z_1b_(1,1) -z_0b_(0,0)-z_1b_(1,0)
| y_1 -y_2 z_0b_(0,2)+z_1b_(1,2) 0
| -y_0 y_1  0                     z_0b_(0,2)+z_1b_(1,2)
| 0    0    -y_1                  -y_2
| 0    0    y_0                   y_1
-------------------------------------------------------
0                       |
-z_0b_(0,0)-z_1b_(1,0) |
-z_0b_(0,1)-z_1b_(1,1) |
-y_3                    |
y_2                     |
*-
-- => the formula in Prop. 4.3 for the pfaffian
--    is correct.

-- Computing C cap det M:
P3=kk[y_0..y_3,b_(0,0)..b_(1,2),Degrees=>{4:1,6:be}]
I1=ideal det sub(M,P3) + sub(C,P3);
I2=radical I1
degree ideal det sub(M,P3), degree I1, degree I2
-- => C intersects det M in 3(be+1) points tangentially
I2_3
-*
y b    + y b    + y b    - y b    - y b    - y b
 1 0,0    2 0,1    3 0,2    0 1,0    1 1,1    2 1,2
*-
(det sub(M,P3)+(I2_3)^2 )% sub(C,P3)==0
-- => formula in proof of Prop. 4.3 is correct
```

```
-- Understanding the small resolutions:
Rw=kk[z_0,z_1,y_0..y_3,b_(0,0)..b_(1,2),w,
    Degrees=>{2:{1,0},4:{0,1},6:{0,be},{0,be+1}}]
L=(sub(M,Rw)+matrix{{0,-w},{w,0}})|matrix{{z_1},{-z_0}};
I=minors(2,L);
eliminate(I,{w});
eliminate(I,{w})==sub(ideal f,Rw)

-- Compute the strict tranforms in X^1_b
-- of the fibers X^2_b -> Y
I1=trim saturate(ideal L_{0}+I);
I2= eliminate(I1,{w});
I1'=trim saturate(ideal L_{1}+I);
I2'= eliminate(I1',{w});
baseLocus=saturate(I2+I2',ideal(basis({1,0},Rw)));
baseLocus==ideal sub(M,Rw)
-- => |O_X^1_b(-1,b+1)| has the (be+1)^3
-- exceptional lines of X^1_b -> Y as the base locus.
```

Running this file completes the proof of all computational claims.

Acknowledgements

Lazić was supported by the DFG-Emmy-Noether-Nachwuchsgruppe "Gute Strukturen in der höherdimensionalen birationalen Geometrie". This work is also a contribution to Project-ID 286237555 – TRR 195 – by the Deutsche Forschungsgemeinschaft (DFG, German Research Foundation).

References

[1] D. Eisenbud, D. Erman, F.-O. Schreyer, M. Stillman, and Y. Kim, *TateOnProducts*, Macaulay2 package available at https://faculty.math.illinois.edu/Macaulay2/doc/Macaulay2/share/doc/Macaulay2/TateOnProducts/html/index.html.

[2] V. Lazić and F.-O. Schreyer, *Macaulay2 supporting file for the paper "Birational geometry of a family of determinantal 3-folds"*, available at https://www.math.uni-sb.de/ag-schreyer/index.php/computeralgebra.

Fachrichtung Mathematik, Campus, Gebäude E2.4, Universität des Saarlandes, 66123 Saarbrücken, Germany
Email address: lazic@math.uni-sb.de

Fachrichtung Mathematik, Campus, Gebäude E2.4, Universität des Saarlandes, 66123 Saarbrücken, Germany
Email address: schreyer@math.uni-sb.de